
Grundlegende Konzepte der Risikotheorie – wie die Gesamtschadenverteilung und Ru-
inwahrscheinlichkeiten – werden in der Praxis nur selten eingesetzt, obwohl sie doch in der
internen Erfolgsrechnung, zum Beispiel bei der Bestimmung interner Solvabilitätskenn-
ziffern oder des für ein Portefeuille erforderlichen Eigenkapitals, durchaus ihren Platz
haben. Diese Konzepte leiden unter dem Vorurteil, daß sie hohen Programmier– und
Rechenaufwand erfordern; und daher werden Untersuchungen, in denen sie vorkommen,
häufig an externe Aktuare, an Rückversicherer und Unternehmensberater vergeben. In
diesem Artikel wollen wir versuchen, das Vorurteil abzubauen und zu zeigen, daß Gesamt-
schadenverteilungen ebenso wie Ruinwahrscheinlichkeiten ohne großen Rechenaufwand
und mit geringem Programmieraufwand zu berechnen sind. Allerdings gehört zur Be-
stimmung unternehmerischer Entscheidungskennzahlen weit mehr als die Berechnung
einer Summenverteilung, und daher wird sich die Auftragslage bei externen Aktuaren
durch unseren Artikel nicht verschlechtern, im Gegenteil: Vielleicht wird das Interesse an
risikotheoretischen Konzepten verstärkt werden.

Wir benutzen zwei Computerprogramme, das Tabellenkalkulationsprogramm Excel
und das Formelmanipulationsprogramm Maple. Unsere Beispiele laufen unter den Versio-
nen Excel 5.0 und 6.0 und unter Maple V, Release 3 und 4, auf PC’s mit Windows.

Um den Text angenehmer zu gestalten, haben wir nur die wichtigsten Formeln aufge-
führt; alles notwendige ist in einem Formelanhang zu finden.

1 Gesamtschadenverteilung unter Excel

Für das Risikomanagement eines Versicherungsportefeuilles benötigt man Information
über die Gesamtsumme aller Schadenzahlungen für Schäden aus einer zukünftigen Zeit-
periode. Dieser Gesamtschaden ist aus heutiger Sicht unbekannt, er wird von vielerlei nicht
vorhersehbaren Ereignissen und Faktoren beeinflußt und daher als Zufallsvariable model-
liert. Dies bedeutet, daß man unter einem geeigneten Modell 1 eine Verteilung berechnet,
welche das zufällige Verhalten des Gesamtschadens beschreibt. Die einfachsten Modelle
für diesen Zweck sind Normalverteilungen und Normal–Power–Approximationen, bei de-
nen eine Gesamtschadenverteilung aus dem Mittelwert, der Varianz und der Schiefe des
Gesamtschadens berechnet wird. Wesentlich näher an der Realität sind Gesamtschaden-

1Albert Einstein soll gesagt haben:”Ein Modell sollte so einfach wie möglich sein, aber nicht zu
einfach”.

verteilungen, die im kollektiven Modell von Lundberg (1903) oder im individuellen Modell
berechnet werden.

Im kollektiven Modell werden die Schadenanzahl und die einzelnen Schadenhöhen
getrennt modelliert, und unter vereinfachenden Annahmen (Unabhängigkeit, Gleichar-
tigkeit der Schadenhöhen) entsteht als Gesamtschadenverteilung eine Summenverteilung
der Form ∞∑

n=0

R{n}Q∗n.

Es ist überraschend, daß bei einer derart einfachen Modellierung Verteilungen entstehen,
für die es – in den meisten Fällen – keine geschlossenen analytischen Formeln gibt und
deren numerische Berechnung wegen der auftretenden Faltungen sehr rechenaufwendig
aussieht. Sonderfälle, in denen analytische Formeln möglich sind, werden im nächsten
Abschnitt behandelt. Hier soll gezeigt werden, wie ein effizientes Berechnungsverfahren
von Panjer (1982) in Excel implementiert werden kann. Dieses Rekursionsverfahren ist
anwendbar für drei Klassen von Schadenzahlverteilungen: Poissonverteilungen, Binomi-
alverteilungen und negative Binomialverteilungen. Wir wollen hier den Fall ganzzahliger
Schadenhöhen betrachten, der z.B. durch Rundung der Schadenzahlungen auf volle DM
entsteht. Die Formel für die Berechnung der Punktwahrscheinlichkeiten von P hat dann
die Form

P{k + 1} =
k+1∑
i=1

(
a+

bi

k + 1

)
Q{i}P{k + 1− i}, k = 0, 1, ...

Die Struktur dieser Formel paßt gut zum Excel–Befehl summenprodukt, bei dem die
Einträge mehrerer Vektoren miteinander multipliziert und die entstehenden Produkte auf-
addiert werden. Allerdings muß man zusätzliche Überlegungen anstellen, da Excel die in
der Funktion summenprodukt eingetragenen Vektoren nur in ihrer natürlichen Indizierung
liest; gibt man beispielsweise den Befehl

summenprodukt(A1 : A10; A10 : A1)

ein, so erscheint als Formel der Ausdruck

summenprodukt(A1 : A10; A1 : A10)

und als Wert die Zahl A12+ ...+A102. Das im folgende beschriebene Spreadsheet berech-
net die Punktwahrscheinlichkeiten P{k} der Gesamtschadenverteilung für k = 0, ..., 300.
Dabei wird jeweils nur der Inhalt der ersten beiden Zeilen angegeben, die Formeln für die
Zeilen 3 bis 301 entstehen automatisch angepaßt durch Kopieren. Die Ausdrücke ParA,

ParB und ParR0 sind Namen für Parameterfelder, die (eventuell auf einer neuen Tabelle)
festgelegt und mit numerischen Werten belegt werden müssen. In Spalte A lassen wir
einen Index von 0 bis 300 laufen. In Spalte B tragen wir die Punktwahrscheinlichkeiten
von Q ein, wobei wir mit Nullen auffüllen. Spalte C enthält dieselben Werte, aber in
umgekehrter Reihenfolge: C1 wird durch die Formel

index(B$1 : B$301; 301− A2)

festgelegt, die restlichen Felder der Spalte werden mit dem Kopierbefehl gefüllt. Die Spalte
D wird mit den Punktwahrscheinlichkeiten von P gefüllt, und zwar D1 mit ParR0, das
Feld D2 mit

summenprodukt(D$1 : D1;

−A$1 : A1 ∗ ParB/A2+ ParA + ParB;

Bereich.Verschieben(C$1 : C1; 301− A2; 0)),

und D3 bis D301 werden wiederum durch Kopieren gefüllt. In unserer Abbildung 1.1
sind die mit diesem Algorithmus berechneten Punktwahrscheinlichkeiten der Summen-
verteilung mit Poissonscher Schadenzahl R = Poi(16) und diskreter Schadenhöhenvertei-
lung Q{2} = Q{9} = 1/2 wiedergegeben.

Abb. 1.1

In der folgenden Abbildung 1.2 wird die Summenverteilung mit R = Poi(4) und
Q{i} = c/i, i = 1, ..., 32 mit 1/c =

∑32
i=1 1/i dargestellt.

Abb. 1.2

Im ersten Beispiel entsteht eine Verteilung, die einer Normalverteilung ähnelt, im
zweiten Fall eine ungewöhnlich schiefe, nicht normale Verteilung. Diese unterschiedlichen
Ergebnisse zeigen die große Anwendungsbreite der oben angegebenen Summenverteilun-
gen; die auf diese Weise modellierten Gesamtschadenverteilungen P haben oft eine völlig
unerwartete Form. Daher ist die Möglichkeit, solche Verteilungen mit einem Standard–
Tool zu berechnen, von großer Bedeutung.

2 Gesamtschadenverteilungen unter Maple

Computeralgebrasysteme [Formelmanipulationsprogramme] haben in den letzten Jahren
eine stürmische Entwicklung erfahren und sind für wirtschaftsmathematische Anwen-
dungen gut geeignet; eine Reihe von Anwendungsbeispielen mit Maple, Mathematica,
Mathcad und Derive [neben Excel] findet man z.B. in Benker (1997). Für das Problem
der Berechnung der Gesamtschadenverteilung [Summenverteilung] bieten sich hier neben
der schon behandelten Panjer–Rekursion vor allem eine ’exakte’ Berechnung z.B. über
die erzeugende Funktion oder eine Anwendung der schnellen Fourier–Transformation an;
versicherungsmathematische Grundlagen dieser Methoden findet man z.B. in Heilmann
(1987) oder Feilmeier und Bertram (1987).
Im folgenden wollen wir wie oben den Fall betrachten, daß die Schadenzahl mit einem
Parameter λ > 0 Poisson–verteilt ist und die Schadenhöhen sich auf positive ganzahlige
Werte konzentrieren. Um den Formalismus überschaubar zu halten, wollen wir der Ein-
fachheit halber in dem hier betrachteten Anwendungsbeispiel annehmen, daß der Träger
der Schadenhöhen aus maximal fünf Werten a, b, c, d, e besteht; es kann aber grundsätzlich
mit jeder beliebigen Zahl von Trägerpunkten a1, a2, . . . , an gearbeitet werden.
Die wahrscheinlichkeitserzeugende Funktion ψ des Gesamtschadens hat in diesem allge-
meinen Fall die Form

ψ(t) = exp


λ


 n∑

j=1

pjt
aj − 1




 , t ≥ 0,

wobei jetzt pj die Wahrscheinlichkeit des Auftretens eines Einzelschadens der Höhe aj

bezeichnet; Es ist also in obiger Notation gerade pj = Q{aj}.
Die Wahrscheinlichkeitsverteilung des Gesamtschadens ist nach obigem durch die Koef-
fizienten der Taylorreihe von ψ um den Nullpunkt gegeben; der wesentliche Befehl in
Maple zur Entwicklung von ψ in ein entsprechendes Taylor–Polynom vom Grad L ist für
unser Beispiel gegeben durch

poly:=taylor(exp(lambda*(p1*t^a+p2*t^b+p3*t^c+p4*t^d+p5*t^e-1)),t=0,L+1):

Um einen sinnvollen Bereich für die numerischen Werte dieser Verteilung zu erhalten,
legen wir vorher den Polynomgrad folgendermaßen fest:

L:=trunc((lambda+sqrt(lambda))*max(a,b,c,d,e));

L liegt damit größenordnungsmäßig im Bereich

(Erwartungswert der Schadenzahl + Standardabweichung)×größter Einzelschaden.
Die errechneten Einzelwahrscheinlichkeiten lassen sich dann folgendermaßen tabellarisch
in Dezimalschreibweise darstellen:

probab:=[seq(evalf(coeff(poly,t,l)),l=0..L)];

Diese drei Befehle genügen im Prinzip, die gewünschten Berechnungen vollständig durchzu-
führen. Das im Anhang wiedergegebene Worksheet enthält darüberhinaus noch einige Be-
fehle zur graphischen Ausgabe der Ergebnisse, die wir hier nicht gesondert kommentieren.
Die folgenden drei Graphiken zeigen das Ergebnis für das obige Beispiel mit R = Poi(16)
und Q{2} = Q{9} = 1/2 [d.h. a = 2, b = 9, c, d, e beliebig und p1 = p2 = 1/2] sowie

R = 10 und verschiedene Werte für a, b, c, d, e und p1, . . . , p5. Man beachte auch hier die
unterschiedlichen Formen der Gesamtschadenverteilung, insbesondere in den letzten bei-
den Graphiken, die sich nur durch die Änderung eines einzigen Trägerpunktes [von b = 6
auf b = 2] unterscheiden.
Der Polynomgrad beträgt im ersten Fall L = 180, in den beiden anderen L = 236; die
Rechenzeit auf einem Notebook PC mit einem Pentium–Prozessor mit 100 MHz Taktfre-
quenz unter Windows 95 betrug im ersten Fall 6.3 sec, im zweiten jeweils 11.5 sec.

Abb. 2.1

Abb. 2.2

Abb. 2.3

Mit Hilfe des ebenfalls im Anhang wiedergegebenen, etwas aufwendigeren Worksheets
zur Fast–Fourier–Transformation lassen sich einjährige Gesamtschadenverteilungen für
Risiko–Todesfallversicherungen berechnen; als Eingabeparameter werden hier eine Ster-
betafel sowie die Policendaten der Versicherten [Alter, Versicherungssumme] benötigt.
Das Worksheet berechnet die Gesamtschadenverteilung ’exakt’ einmal nach der schon
besprochenen Methode der erzeugenden Funktion, zum anderen mit der Fast–Fourier–
Methode [dazu gibt es automatisierte Befehle unter Maple] sowie aproximativ nach der
Panjer–Methode, wobei die binomialverteilten Zufallsgrößen Xi mit den Werten 0 und
1 für ’Überlebensfall’ und ’Todesfall’ für die Versicherten i jeweils durch eine Poisson–
verteilte Zufallsgröße Yi mit demselben Erwartungswert λi = E(Xi) aproximiert werden.
Wegen der i.a. geringen einjährigen Sterbewahrscheinlichkeit ist dieses Vorgehen durch
den Poisson’schen Grenzwertsatz [’Gesetz der seltenen Ereignisse’] gerechtfertigt. Die
Einzelschadenhöhen werden dabei wie üblich über Vielfache des größten gemeinsamen
Teilers diskretisiert.
Die folgende Graphik zeigt das Ergebnis einer Rechnung für ein Kollektiv von 50 Ver-
sicherten mit Altern zwischen 20 und 60 Jahren sowie unterschiedlichen Versicherungssum-
men zwischen DM 5000,- und DM 410000,- [ggT = DM 5000,-], dargestellt in logarith-
mischem Maßstab. Das zur erzeugenden Funktion gehörige Polynom ist hier vom Grad
631!

Abb. 2.4

Man beachte, daß sich die numerischen Werte nach der (hier deutlich schnelleren)
Methode der erzeugenden Funktion und der (hier langsameren) Fast–Fourier–Transforma-
tion praktisch nicht unterscheiden, und die Panjer–Poisson–Approximation im Vergle-
ich dazu lediglich geringfügige Fehler von unter 10−5 für die Einzelwahrscheinlichkeiten
aufweist.

Eine vollständige Programmsamlung zur Versicherungsmathematik mit Maple sowie Bei-
spieldateien sind in Pfeifer (1997) enthalten und können vom Autor direkt bezogen wer-
den.

3 Ruinwahrscheinlichkeiten unter Maple

Wir betrachten die Ruinwahrscheinlichkeit ψ(s) im klassischen Risikoprozeß (also mit un-
endlichem Planungshorizont und ohne Zinsen), die vom Startkapital s, der Schadenhöhen-
verteilung Q, der Prämienrate c und der Poisson–Intensität λ abhängt. Für exponen-
tialverteilte Schadenhöhen mit Mittelwert µ, also für Q = Exp(1/µ), kann man ψ(s)
explizit darstellen:

ψ(s) =
λµ

c
exp

(
− s

µ

(
1− λµ

c

))
.

Es gibt eine große Klasse von Verteilungen, für die man eine explizite Darstellung für
ψ(s) mit Maple erhalten kann: die sogenannte Klasse der Phasentyp–Verteilungen. Sie
sind von Neuts (1975) entdeckt worden, und Asmussen und Rolski (1992) haben sie in der
Ruintheorie eingesetzt. Diese Klasse umfaßt alle Exponentialverteilungen, alle Faltungen
von Exponentialverteilungen und alle endlichen Mischungen von Exponentialverteilun-
gen. Die Klasse ist abgeschlossen unter Faltungen und unter endlichen Mischungen, und
sie liegt dicht in der Menge aller möglichen Schadenhöhenverteilungen bezüglich der
Verteilungskonvergenz. Mit dieser Menge von Schadenhöhenverteilungen hat man also
eine große Modellierungsvielfalt.

Phasentyp–Verteilungen (der Stufe I) entstehen bei Markov–Ketten X(t), t ≥ 0, auf
dem Zustandsraum {0, 1, ..., I},bei denen der Zustand 0 absorbierend ist. Solche Markov–
Ketten werden charakterisiert durch einen Startvektor p und eine Intensitätsmatrix B :

P{X(0) = i} = pi, i = 1, ..., I,

P{X(t+ h) = j|X(t) = i} = bijh+ o(h), i, j = 1, ..., I, i �= j,

P{X(t+ h) = i|X(t) = i} = 1− biih+ o(h), i = 1, ..., I.

Die Phasentyp–Verteilung mit den Parametern p und B ist die Verteilung der Wartezeit
T bis zur Absorption in der zugehörigen Markov–Kette:

T = inf{t : X(t) = 0}.
Bei I = 2 und

B =

( −a a
0 −b

)

erhält man für p = (0, 1) die Exponentialverteilung Exp(b), und für p = (1, 0) entsteht die
Faltung von Exp(a) und Exp(b). Die Verteilung Q mit p, B hat Tailwahrscheinlichkeiten

Q(t,∞) = p′ exp(tB)e,

wobei e der Vektor ist, dessen Komponenten alle 1 sind. Phasentyp–Verteilungen haben
eine rationale Laplace–Transformierte, und daher kann die Gammaverteilung mit Para-
metern 1/2, 1/2 nicht phasentyp sein.

Ist Q die Phasentyp–Verteilung mit Parametern p, B, dann ist ψ(s) Tailwahrschein-
lichkeit einer Phasentyp–Verteilung:

ψ(s) =
λµ

c
p′∗ exp(tB∗)e,

wobei

p∗ = −1
µ
(B′)−1

p und (B∗)ij = bij +
λµ

c
pjbi0, i, j = 1, ..., I.

Diese Formel kann man nun sehr einfach mit Maple berechnen: im Modul linalg sind
nämlich Exponentiale von Matrizen implementiert. Die Zeilen

with(linalg):

B:=matrix(2,2,[-a,a,0,-b]);

C:=exponential(B,t);

erzeugen beispielsweise das Ergebnis

C :=

[
e−at a

a−b

(
e−at − e−bt

)
0 e−bt

]
.

Die nächste Grafik zeigt die Ableitung −ψ′(s) der Überlebenswahrscheinlichkeit 1−ψ(s)
für die Schadenhöhenverteilung, die die Mischung zweier Gammaverteilungen ist:

Q = 0.5 Gamma(2; 2) + 0.5 Gamma(2; 0.01),

bei vorgegebenem ψ(0) = 2/3. Man sieht, daß eine Erhöhung des Eigenkapitals bis auf
etwa 60 Einheiten mehr Sinn macht als eine noch weitergehende Erhöhung.

Abb. 3.1

4 Anhang
Die Gesamtschadenverteilung (Summenverteilung)

Die Summenverteilung, welche im ersten Abschnitt erwähnt wurde, entsteht folgendermaßen: Der Gesamt-
schaden eines Portefeuilles wird modelliert durch die Schadenanzahl N und die einzelnen Schadenhöhen
X1, X2, ...: S = X1 + ... + XN . Unter der Annahme, daß die Schadenanzahl und die Schadenhöhen
stochastisch unabhängig sind mit R als Schadenzahlverteilung und Q als Schadenhöhenverteilung, entsteht
die Verteilung

∞∑
n=0

R{n}Q∗n

für den Gesamtschaden S. Q∗n ist die n−fache Faltung, und Q∗0 ist die in Null konzentrierte Verteilung,
wobei Q∗0{0} = 1.

MAPLE–Worksheet zur Gesamtschadenverteilung

Mit diesem Worksheet kann die Gesamtschadenverteilung im kollektiven Modell berechnet und graphisch
dargestellt werden, wobei als Schadenzahlverteilung eine Poisson–Verteilung mit Parameter lambda>0
und als Schadenhöhenverteilung eine auf maximal fünf vorgebbare Punkte konzentrierte Verteilung betra-
chtet werden. Die exakte Berechnung der Gesamtschadenwahrscheinlichkeiten erfolgt über die wahrschein-
lichkeitserzeugende Funktion.
>restart: E:=exp(1):
>with(plots):

Eingabe von lambda (Poisson–Parameter):
>lambda:=10;

Eingabe der Trägerpunkte a,b,c,d,e>0:
>a:=1; b:=6; c:=12; d:=13; e:=18;

Eingabe der Wahrscheinlichkeiten p1,p2,p3,p4 für a,b,c,d:
>p1:=0,1; p2:=0.35; p3:=0.05; p4:=0.2; p5:=1-p1-p2-p3-p4;

Berechnung der wahrscheinlichkeitserzeugenden Funktion:
>L:=trunc((lambda+sqrt(lambda))*max(a,b,c,d,e));

>poly:=taylor(exp(lambda*(p1*t^a+p2*t^b+p3*t^c +p4*t^d+p5*t^e-1)),t=0,L+1):
>probab:=[seq(evalf(coeff(poly,t,l)),l=0..L)];

Graphische Darstellung:
>Index:=[seq(r,r=0..L)]:
>Graph:=zip((u,v)->[u,v],Index,probab):
>plot(Graph,style=LINE,title=cat(‘Gesamtschaden: lambda=‘,lambda,‘,Träger=
[‘,a,‘,‘,b,‘,‘,c,‘,‘,d,‘,‘,e,‘], p=[‘,convert(p1,string),‘ ‘,convert(p2,string),‘ ‘,
convert(p3,string),‘ ‘,convert(p4,string),‘ ‘,convert(p5,string),‘]‘));

MAPLE–Worksheet zur Fast–Fourier–Transformation

Mit diesem Worksheet können einjährige Schadensummenverteilungen aus Lebensversicherungsporte-
feulles exakt, mit der Poisson–Panjer–Approximation und der FFT–Methode approximativ berechnet und
graphisch dargestellt werden. Als Eingabedateien werden eine aktuelle Sterbetafel [einspaltig, ASCII–
Format] sowie eine Datei mit den Angaben: Eintrittsalter x und Versicherungssumme VS [in ganzen
Zahlen; zweispaltig, ASCII–Format] benötigt.
>restart: E:=exp(1):

File mit aktueller Sterbetafel:
>file:=‘astm91.int‘;

File mit Versicherten–Daten:
>file2:=‘daten2.pan‘;

Laden verschiedener MAPLE–Sub–Pakete (Lineare Algebra, Zeichnen):
>with(linalg):
>with(plots):

Einlesen der Daten (Sterblichkeitsintensitäten µ):
>readlib(readdata):
>readlib(FFT):
>mu:=readdata(file):
>omega:=vectdim(mu)-1;
[Endalter]

Einlesen der Versicherten–Daten: x,VS:
>Daten:=readdata(file2,integer,2):
>Anzahl:=vectdim(Daten);
[Anzahl der Fälle]

>Daten:=array(Daten);
>Alter:=seq(Daten[j,1],j=1..Anzahl):
[Eintrittsalter]

>VS:=seq(Daten[j,2],j=1..Anzahl):
[Versicherungssummen]

>Delta:=igcd(VS);
[Schrittweite]

>m:=[seq(VS[r]/Delta,r=1..Anzahl)]:
>s:=sum(m[i],i=1..Anzahl):

Exakte Berechnung der Wahrscheinlichkeiten mit der erzeugenden Funktion:
>erzpoly:=expand(product((1+mu[Alter[i]+1]*(t^m[i]-1)),i=1..Anzahl)):
>probab:=[seq(coeff(erzpoly,t,l),l=0..s)]:
>A:=0:
>while(probab[A+1]>0.0000001 or probab[A+1]=0) do A:=eval(A)+1: prob[A]:=probab[A]: od:

>pr:=[seq(prob[i],i=1..A)]:
[Anfängliche exakte Wahrscheinlichkeiten oberhalb 0.0000001]

Poisson–Approximation:
>lambda:=sum(mu[Alter[u]],u=1..Anzahl):
>ps[0]:=evalf(E^(-lambda)):
>for k from 1 to A do p[k]:=0: od:
>for k from 1 to Anzahl do p[m[k]]:=p[m[k]]+mu[Alter[k]]: od:

Panjer–Rekursion:
>for k from 0 to A-1 do ps[k+1]:=1/(k+1)*sum((k+1-v)*p[k+1-v]*ps[v], v=0..k) od:
>Index:=[seq(z*Delta,z=0..A-1)]:
>exakt=pr;
>Approx:=[seq(ps[k],k=0..A-1)];

FFT–Approximation:
>L:=trunc(evalf(log[2](A)))+1;
>M:=2^L;
>z:=convert([seq(1,r=1..M)],array):
>for i from 1 to Anzahl do x:=convert([seq(0,r=1..M)],array):
x[1]:=1-mu[Alter[i]]: x[m[i]+1]:=mu[Alter[i]]: y:=convert([seq(0,r=1..M)],array):
FFT(L,x,y): for j from 1 to M do z[j]:=(x[j]+y[j]*I)*z[j]:od: od: for j from 1 to M do
x[j]:=Re(z[j]): y[j]:=Im(z[j]): od:

>iFFT(L,x,y):
>exakt:=pr;
>Approx:=[seq(ps[k],k=0..A-1)];
>FFT:=[seq(x[k],k=1..A)];

Graphische Darstellung: exakt = —, FFT = o, Panjer = +:
[Aus Gründen der Darstellbarkeit im logarithmischen Maßstab werden die Wahrscheinlichkeiten mit Wert
0 auf den Wert 10ˆ(-8) abgebildet]
>Index:=[seq(Delta*k,k=0..A-1)]:
>Graph:=zip((u,v)->[u,max(v-10^(-8),10^(-8))],Index,FFT):
>Graph2:=zip((u,v)->[u,max(v-10^(-8),10^(-8))],Index,exakt):
>Graph3:=zip((u,v)->[u,max(v-10^(-8),10^(-8))],Index,Approx):
>pic1:=logplot(Graph,style=point,symbol=CIRCLE, title=cat(‘Einjährige
Gesamtschadenverteilung, exakt: ---, FFT: o, Panjer: +, ‘,Anzahl,‘ Fälle‘),color=navy):
pic2:=logplot(Graph2,style=PATCH,color=red):
pic3:=logplot(Graph3,style=point,symbol=CROSS,color=violet):
display([pic1,pic2,pic3]);

MAPLE–Worksheet zur Phasentyp–Verteilung

Das Maple–Programm, welches Abb. 3.1 produziert, hat folgende Form:

>restart:
>with(linalg):
>B:=matrix(4,4,[-2,2,0,0,0,-2,0,0,0,0,-0.01,0.01,0,0,0,-0.01]);
>pi:=vector(4,[1/2,0,1/2,0]);e:=vector(4,[1,1,1,1]);
>pist:=evalm(inverse(transpose(B))&*pi);
>mean:=evalm(transpose(pist)&*e);
>pist1:=evalm(pist/mean);
>Lambda:=vector(4,[0,2.,0,0.01]);
>p:=2/3;
>C:=evalm(Lambda&*transpose(pist1));
>Bst:=evalm(B+p*C);
>g1:=p*(evalm(transpose(pist1)&*exponential(Bst,t)&*e));
>g2:=-diff(g1,t);
>plot(g2,t=2..150);

